
第３章　プログラム例

I/O制御・マルチタスク

MPC の構成と装置概要

図のよ うなP&Pのプログラムを考えます。このP&PはコンベアーＡでパレッ トに乗って搬送されているワー

クをス ト ッパーで止め、エアーシ リ ンダーで構成されたP&PでコンベアーＢのパレッ トへ乗せかえる という

ものです。入力はフォ トマイ ク ロセンサーまたはリードスイ ッチ、出力モーターはＳ Ｓ Ｒを介してON/OFF

を行いそれ以外はソレ ノ イ ドバルブを直接駆動する とい う設定です。この例でのMPCのハード構成はMPC-

684、 MIP-048、 MOP-048、 RACK-N6を各1台ずつ、 アドレス設定は出荷時のまま使用します。

I/Oマップ

入力 （MIP-048） 出荷時の設定ではポート番号は192から始ま り ます。

出力 （MOP-048） 出荷時の設定ではポート番号は0から始ま り ます。

メモ リー I/O メモ リー I/Oは-1から-8192です。

コンベアＡコンベアＢ

SW196 ワーク到達

ON3 ワークストッパON4
ワークストッパ ワーク

SW197 ワーク到達

SW193 SW192

上昇端
SW194( 原点 )

SW195
下降端

【P&P図】

ポート 接続機器 名　称 機　能

１９２ シリンダーセンサー ｋｏｔａｎ Ｐ＆Ｐ後退端

１９３ シリンダーセンサー ｚｅｎｔａｎ Ｐ＆Ｐ前進端

１９４ シリンダーセンサー ｕｅ Ｐ＆Ｐ上昇端

１９５ シリンダーセンサー ｓｈｉｔａ Ｐ＆Ｐ下降端

１９６ フォトマイクロセンサー ｗｏｒｋ１ コンベアーＡワーク到達

１９７ フォトマイクロセンサー ｗｏｒｋ２ コンベアーＢワーク到達

ポート 接続機器 名　称 機　能

０ ソレノイドバルブ ｚｅｎｓｈｉｎ Ｐ＆Ｐ前進

１ ソレノイドバルブ ｋａｋｏｕ Ｐ＆Ｐ下降

２ ソレノイドバルブ ｃｈａｃｋ メカチャック閉

３ ソレノイドバルブ ｓｔｏｐｐｅｒ１ コンベアーＡストッパー

４ ソレノイドバルブ ｓｔｏｐｐｅｒ２ コンベアーＢストッパー

５ ＳＳＲ ｍｏｔｏｒ１ コンベアーＡモーター

６ ＳＳＲ ｍｏｔｏｒ２ コンベアーＢモーター

ポート 接続機能 名　称 機　能

－１ メモリーＩ／Ｏ ａ＿ｒｅａｄｙ コンベアーＡＲＥＡＤＹ

－２ メモリーＩ／Ｏ ｂ＿ｒｅａｄｙ コンベアーＢＲＥＡＤＹ

この装置の動作は単純で Ａから Ｂ への移し替えの仕事は1つのシーケンス と してプログラム出来るかも知

れません。しかし、実際の多くの装置はとても複雑で、タ ク ト タ イムなどを考える と適当なユニッ ト単位に

分割してそれぞれが独立して動作するよ うに制御するのが合理的です。 例えばこの装置の場合コンベアー

Ａの部分、 P&Pの部分、 コンベアー Ｂの部分という 3つブロ ッ クに分けてプログラムします。 このブロ ッ ク

をタスク と言い、MPC-684では複数のタスクを同時に実行するのでマルチタスク と言います。しかしそれぞ

れのタスクが勝手に動作していたのでは装置と しての目的を達成するこ とは出来ません。 そこでタスク同

士の連絡を行う必要があ り これをインターロ ッ ク と言います。 インターロ ッ クは、 メモ リーI/Oも し くは変

数で行います。 メモ リー I/Oとは通常装置間のインターロ ッ クを物理的なI/Oで行う よ うにタスク間のイン

ターロ ッ クを行うための内部的なI/Oです。メモ リーI/Oは物理的なI/Oと同様に1つのビッ トで1(ON)または

0(OFF)の2つの状態を表します。 そしてMPC-684ではON/OFF/SWなどのコマンドに引き数を負の整数と し

て与える事によ り メモ リーI/Oと して、物理的なI/Oと同様に扱う事ができます。1つのメモ リーI/OではON/

OFFのどちらしか無いのに対し、 変数を用いる と複数の情報を伝えるこ とができます。 MPC-684の変数は

4byte整数ですから、 1つの変数で約±20億の状態を表すこ とができます。

I/O のシンボル化

プログラム中の入出力ポートのパラ メーターと して I/O のポートナンバーを整数でそのまま記述する事が

できますが、シンボルと して扱う と見た目にも分かりやすいプログラムにな り ます。次のよ うにCONSTコマ

ンドで定数をシンボル化します。

CONST  sol1 0
CONST  sw1  192

これでON 0とON sol1,WAIT SW(192)=0はWAIT SW(sw1)=0と同義になり ます。

（P&P図） の装置のI/O定義

10 'I/O定義
20 '入力
30 CONST kotan 192
40 CONST zentan 193
50 CONST ue 194
60 CONST shita 195
70 CONST work1 196
80 CONST work2 197
90 '出力
100 CONST zenshin 0
110 CONST kakou 1
120 CONST chack 2
130 CONST stopper1 3
140 CONST stopper2 4
150 CONST motor1 5
160 CONST motor2 6
170 'メモリー I/O
180 CONST a_ready -1
190 CONST b_ready -2
200 'I/O初期化
210 SETIO

タスクの生成

それぞれのユニッ ト を1～31のどのタスクで実行するかを宣言します。0はメ インのタスクで、プログラムが

実行される と最初にこのタスクが動きます。 この例ではP&Pはタスク0で、 コンベアーＡはタスク1、 コンベ

アー Ｂはタスク2で動かします。

220 'タスク実行
230 FORK 1 *conv_a ’タスク1でコンベアーＡを実行
240 FORK 2 *conv_b ’タスク2でコンベアーＢを実行

P&P のタスク

P&Pはコンベアー Ａ からのREADY を待ちワークをピッ クアップしコンベアー Ｂ 側に搬送します。 コンベ

アー Ｂの準備が出来たと ころでワークを置きます。 このタスクはメ インタスク （タスク0） で実行します。

1000 *PANDP
1010 '
1020 DO
1030 WAIT SW(a_ready)==1 ’コンベアＡREADY待ち
1040 '
1050 ON kakou ’ワークピックアップ
1060 WAIT SW(shita)==1
1070 ON chack
1080 TIME 500
1090 OFF kakou
1100 WAIT SW(ue)==1
1110 '
1120 OFF a_ready ’ピックアップ完了－＞コンベアＡ
1130 '
1140 ON zenshin ’P&P前進
1150 WAIT SW(zentan)==1
1160 '
1170 WAIT SW(b_ready)==1 ’コンベアーＢ準備完了待ち
1180 '
1190 ON kakou ’ワークプレース
1200 WAIT SW(shita)==1
1210 OFF chack
1220 TIME 300
1230 OFF kakou
1240 WAIT SW(ue)==1
1250 '
1260 OFF b_ready ’ワークプレース完了－＞コンベアー
1270 '
1280 OFF zenshin ’P&P後退
1290 WAIT SW(zentan)==1
1300 '
1310 LOOP

コンベアーＡのタスク

コンベアー Ａはス ト ッパーを上げmotor1を駆動してパレッ トが到達するのを待ちます。 パレッ トが到達す

る とP&PにREADYを出し、 P&Pはワークを持ち上げる とREADYをOFFします。

2000 *conv_a
2010 ON motor1
2020 DO
2030 ON stopper1
2040 WAIT SW(work)==1
2050 TIME 500
2060 OFF motor1
2070 '
2080 ON a_ready ’ワークが来たョ
2090 WAIT SW(a_ready)==0 ’P&P動作完了待ち
2100 '
2110 OFF stopper1
2120 ON motor1
2130 WAIT SW(work1)==0
2140 TIME 500
2150 LOOP

コンベアーＢのタスク

コンベアーＢはモーターを回しス ト ッパーを上げ空パレッ トが到達するのを待ち、P&PへREADYを出しP&P

はワークを置く とREADYをOFFします。

3000 *conv_b
3010 ON motor2
3020 DO
3030 ON stopper2
3040 WAIT SW(work2)==1
3050 TIME 500
3060 OFF motor2
3070 '
3080 ON b_ready ’パレット準備完了
3090 WAIT SW(b_ready)==1 ’ワークを置いたョ
3100 '
3110 ON motor2
3120 OFF stopper2
3130 WAIT SW(work2)==0
3140 TIME 500
3150 LOOP

装置全体のプログラム

前記プログラムを入力してRENUMする と次のLISTにな り ます。

10   'I/O定義
20   '入力
30    CONST kotan 192
40    CONST zentan 193
50    CONST ue 194
60    CONST shita 195
70    CONST work1 196
80    CONST work2 197
90   '出力
100   CONST zenshin 0 
110   CONST kakou 1
120   CONST chack 2
130   CONST stopper1 3
140   CONST stopper2 4
150   CONST motor1 5
160   CONST motor2 6
170  'メモリー I/O
180   CONST a_ready -1
190   CONST b_ready -2
200  'I/O初期化
210   SETIO
220  'タスク実行
230   FORK 1 *conv_a ’タスク1でコンベアーＡを実行
240   FORK 2 *conv_b ’タスク2でコンベアーＢを実行
250  *PANDP
260  '
270   DO
280     WAIT SW(a_ready)==1 ’コンベアＡREADY待ち
290  '
300     ON kakou ’ワークピックアップ
310     WAIT SW(shita)==1
320      ON chack
330      TIME 500
340      OFF kakou
350      WAIT SW(ue)==1
360  '
370      OFF a_ready ’ピックアップ完了－＞コンベアＡ
380  '
390      ON zenshin ’P&P前進

400      WAIT SW(zentan)==1
410  '
420    WAIT SW(b_ready)==1 ’コンベアーＢ準備完了待ち
430  '
440    ON kakou ’ワークプレース
450    WAIT SW(shita)==1
460    OFF chack
470    TIME 300
480    OFF kakou
490    WAIT sw(ue)==1
500  '
510    OFF b_ready ’ワークプレース完了－＞コンベアーＢ
520  '
530    OFF zenshin ’P&P後退
540    WAIT SW(zentan)==1
550  '
560    LOOP
570  *conv_a
580    ON motor1
590    DO
600    ON stopper1
610    WAIT SW(work1)==1
620    TIME 500
630    OFF motor1
640  '
650    ON a_ready ’ワークが来たョ
660    WAIT SW(a_ready)==0 ’P&P動作完了待ち
670  '
680    OFF stopper1
690    ON motor1
700    WAIT SW(work1)==0
710    TIME 500
720    LOOP
730  *conv_b
740    ON motor2
750    DO
760      ON stopper2
770      WAIT SW(work2)==1
780      TIME 500
790      OFF motor2
800  '
810    ON b_ready ’パレット準備完了
820    WAIT SW(b_ready)==0 ’ワークを置いたョ
830  '
840    ON motor2
850    OFF stopper2
860    WAIT SW(work2)==0
870    TIME 500
880    LOOP

タスクの制御

タスクの交通整理（セマフォ）

タスク間のインターロ ッ クが必要なケースと して先の例でのタスクの同期と、も う 1つ、一度に1つのタスク

しか使用できない共通資源 （ク リ ティカルソース） に対し排他処理を行う場合があ り ます。 例えば、 1つの

RS-232 ポートやリエン ト ラン ト出来ないサブルーチンを複数のタスクで使用した り、 重なる動作エリ アを

持つ2台のロボッ ト （図） の干渉防止などがあ り ます。 本来、 前者のよ うな一人の聞き手に対して一度に何

人もの人が同時に話しかけるよ う なプログラムは避けるべきです。 聖徳太子のよ うに器用に聞き分けるこ

とはできないのです。1つのデバイスは1つのタスクで管理するこ とが原則ですが、それができないと きは複

数のタスクが同時にアクセスしないよ うにしなければな り ません。 後者はタスクのアクセスする対象は異

なり ますが、ポイン トや空間が共通資源と して考えられます。このよ う なタスクの交通整理をセマフォ と言

い、 それを合理的に確実に行う方法と して、 セマフォ関数 （RSV(n),RLS(n)） があ り ます。

1. プログラムはコマンド リ ファレンスのRSV(n)の項のプログラムです。 このプログラムは2つのタスクが勝

手に走ってRS-232のCH1を同時にアクセスしてしまい出力が混乱しています。 も し も これが2台のロボッ

ト なら衝突して大きな事故になってしまいます。

2. メモ リー I/OとON/OFF/SW(n)を組み合わせてテス ト &セッ ト を行う こ とによ り タスクの衝突を防ご う と

しています。しかし この方法ではメモ リーI/Oのテス ト とセッ トが別々のコマンドによって行われている

ため時間差が生じてタイ ミ ングによってはやはり混乱が生じる恐れがあ り ます。

3. メモ リー I/Oをセマフォ関数を使ってテス ト &セッ ト しています。 RSV(-1)でメモ リー I/Oの-1の状態を確

認し 0であればそれに 1 をセッ トする事によ りセマフォの獲得をして PRINT文を実行し、 RLS(-1) でセマ

フォを解放します。 この様にRSVは2.） のSWとONによるテス ト &セッ ト を1つ関数で行う事によ り確実な

排他処理を実現できます。

4. RS-232へ出力する専用のタスクを設けたものです。こ こでも メモ リーI/Oを使ってタスクの情報交換をし

ています。

1.2つのタスクが勝手に動作 （交通整理が行われていない）

’もしもロボットなら
10     time0=timer 　　↓
20     time=timer
30   FORK 1 *aho

ロボットＢ

ロボットＡ

コンベア

パレット

動作エリア

重なる動作エリア

Ｐ（ｎ）

MPG#1
タスク 0

MPG#2
タスク 0

【2台のロボッ ト図】

＊コンベアー上のワークを2台のロボッ トのどちらかでパレッ トに搬送する。

　片方がP(n)へアクセス中、 他方は絶対にそのエリ アに進入してはならない。

40   DO
50     WAIT time<>timer
60     PRINT "abcdefg" "ABCDEFGH" ’ロボットＡがP(n)へ動作
70       time=timer
80   LOOP
90 *aho
100  DO
110    WAIT time0<>timer
120    PRINT "123456" "7890123" ’ロボットＢがP(n)へ動作
130      time0=timer
140  LOOP
#run
123456789abcdefgABC0123 ’ガーン衝突！エライコッチャ　
DEFGH
abcdefgAB1234567890CDEFGH
123
12345678901abcdefgA23
BCDEDGH

（timerは予約変数です）

2.ON/OFF/SW(n)による交通整理 （悪い例 ・交通整理できません）

10      time0=timer
20      time=timer
30    OFF -1
40    FORK 1 *aho
50    DO
60      WAIT time<>timer
70      WAIT SW(-1)==0
80      ON -1
90      PRINT "abcdefg" "ABCDEFGH"
100     OFF -1
110       time=timer
120   LOOP
130 *aho
140   DO
150     WAIT time0<>timer
160     WAIT SW(-1)==0
170     ON -1
180     PRINT "123456" "7890123"
190     OFF -1
200       time0=timer
210    LOOP
#run
abcdefgABCDEFGH ’たまたまうまくいった。デモ．．．
1234567890123
abcdefgABCDEFGH
1234567890123

3.セマフォ関数を用いた交通整理 （良い例）

10      time0=timer
20      time=timer
30    FORK 1 *aho
40    DO
50      WAIT time<>timer
60      WAIT RSV(-1)==0 ’セマフォ確認
70      PRINT "abcdefg" "ABCDEFGH" ’ロボットＡ動作
80        dummy=RLS(-1) ’セマフォ解除
90        time=timer
100   LOOP
110 *aho
120   DO
130     WAIT time0<>timer
140     WAIT RSV(-1)==0 ’セマフォ確認
150     PRINT "123456" "7890123" ’ロボットＢ動作

160       dummy=RLS(-1) ’セマフォ解除
170       time0=timer
180   LOOP
#run
abcdefgABCDEFGH ’ロボットＡ動作
1234567890123 ’ロボットＢ動作
abcdefgABCDEFGH 　メデタシメデタシ
1234567890123

4.RS-232への出力を1本化 （良い例）

10      time0=timer
20      time=timer
30    FORK 1 *aho
40    FORK 2 *baka
50    DO
60      IF SW(-1)==1 THEN : GOSUB *case1 : OFF -1 : END_IF
70      IF SW(-2)==1 THEN : GOSUB *case2 : OFF -2 : END IF
80    LOOP
90  *case1
100   PRINT "abcdefg" "ACDEFGH"
110   RETURN
120 *case2
130   PRINT "123456" "78901234"
140   RETURN
150 *aho
160   DO
170     WAIT time<>timer
180     ON -1
190     WAIT SW(-1)==0
200       time=timer
210   LOOP
220 *baka
230   DO
240     WAIT time0<>timer
250     ON -2
260     WAIT SW(-2)==0
270       time0=timer
280   LOOP
#run
12345678901234
abcdefgABCDEFGH
12345678901234
abcdefgABCDEFGH

タスクの FORK、QUIT、PAUSE

インタプ リ タはタスク0を親のタスク と して実行し、FORKコマンドで子タスクを実行します。子タスクには

優先順位や連続性はあ り ません。昇順、降順、中飛びは関係あ り ません。また、子タスク同士のFORK、PAUS

E、 QUITもできます。 たとえば、 自動モード／単動モードの切り替えを行うのに常に自動と単動の2つのタ

スクを走らせておく必要はあ り ません。自動、単動それぞれルーチンを用意しておき自動モードでは単動の

ルーチンをQUITし、 自動のルーチンをFORKします。 単動モードではその逆をします。 このよ うに、 1つの

タスクでもFORKするプログラムを差し替えるこ とによ り複数の機能を実現する事ができます。

*MAIN
  DO
    WAIT SW(start)==0
    WAIT SW(start)==1 ’スタートスイッチ入力待ち
    IF SW(mode)==1 THEN
      FORK 1 *AUTO ’modeスイッチがONならばAUTOモード
      ELSE
      FORK 1 *MANUAL
    END_IF
    WAIT SW(stop)==1 ’ストップスイッチ入力待ち
    QUIT 1 ’タスク1をQUIT

    WAIT TASK(1)==-2
  LOOP
'*********************
*AUTO ’自動運転のプログラム
   DO
     JUMP P(1)
     ON 0
     JUMP P(2)
     OFF 0
   LOOP
'*********************
*MANUAL ’単動運転のプログラム
   DO
     JUMP P(1)
     ON 0
     WAIT SW(tando)==0
     WAIT SW(tando)==1
   LOOP

パルス発生について

ボードアドレスとマルチタスク

MPG-314はDSW1を0-9に設定するこ とで、ひとつのシステムで10枚まで使用するこ とができます。そのと き

のボードのアドレスはDSW番号にしたがって、 &H400～&H490とな り ます。

また、 各タスク毎にPGコマンドで使用するPGを選択するこ とができます。

PG &H400    /* DSW1=0のボードを使用する
PG &H420 1  /* DSW1=2のボードをタスク1に割り当てる

最初の確認

MPC-684はMPG-314を自動認識して初期化します。 POWER ON後LED17,18(緑)が点灯になり ます。

LED17が点灯しない->ラ ッ クに正し く挿入されていない

LED18が点灯しない->MPG-314にDC24Vが接続されていない

LEDの点灯を確認したらパルス発生の確認をします。

DSWの値にしたがってPGコマンドを実行し、 T<Enter>します。 この時J2,J3,J4はすべてコネクタをと りはず

して ください。

#PG &h410 /*DSWが1の場合
#t
PG[0,410] X= 0 Y= 0 U= 0 Z= 0 dx= 500 dy= 500 du= 500 dz= 500

前記の表示を確認したら、x,y,z,uキーを押してみます。LED1～8のいずれかが点灯しパルス発生を確認でき

ます。 SHIFTキーを押すとCCW側にパルス発生します。

これによ り簡単なインチング動作、パルス発生確認ができます。このコマンドを停止するにはQを押します。

※注意

パルスが出ない場合。

J2が接続され、 X-LMT～Z-LMTがONになっている。

J2の リ ミ ッ ト入力は、 無条件に有効とな り ます。 変更できるのはON/OFFの論理だけです。

dx=500でない。

0～3の数字キィーを押すと変更できます。 コマンド"SET"でこの値を変更できます。

簡単なパルス発生プログラム

必要なIOを接続してパルス発生をします。 次に必要なコマンドを使用した例を示します。

それぞれのコマンドの詳細はコマンド リ ファレンスを参照して ください。次はX軸に対してです。Y,Z,Uの各

軸に対しては、 Y_A,Z_A,U_Aを指定します。

10 PG &H410
20 ACCEL X_A -25 4000 400 100 /* 軸指定 S字の割合(25%) 

/* 4000PPSMAX 400(加速距離) 100(自起動)
25 RANGE X_A 50000 -50000 /*動作範囲設定 ソフトリミット
30 INSET_314 X_A SLMT_ON|LMT_ON /*MPG-314動作フラグ設定

/* "SLMT_ON" RANGEで設定した値を有効。
/* "LMT_ON" リミット入力をONで有効に設定。

40 STPS X_A 0 /*現在位置変更
50 RMVS X_A -10000 /*相対移動(ｲﾝｸﾘﾒﾝﾀﾙ)でCCWで10000パルス発生。
60 WAIT RR(X_A)==0 /*パルス発生の終了待ちです。

INSET_314 コマンドの役割は、 ビッ ト定数という MPC-684内部で定義された値をMPG-314の内部レジスタ

WR2に設定するもので、 リ ミ ッ ト論理、 インポス論理、 ソフ ト リ ミ ッ トの有効、アラーム入力の論理を決定

します。 一度設定する と、 次のINSET_314コマンド まで保持されます。

このコマンドは'上書きコマンド'です。 このため、 以前の設定値はク リ アされます。
3-10



※注意 INSET_314について

リ ミ ッ ト入力を無効にするこ とはできません。 ON/OFFを選ぶだけです。 

LMT_ON /* リ ミ ッ ト入力ONで検知

LMT_OFF /* リ ミ ッ ト入力OFFで検知

ソフ ト リ ミ ッ トは有効 ・無効の決定のみです。

SLMT_ON /* ソフ ト リ ミ ッ ト有効

SLMT_OFF /* ソフ ト リ ミ ッ ト無効

インポズ設定,アラーム設定はON/OFFを選択する と自動的に有効になり ます。

INP_ON /* インポズ入力ONで検知

INP_OFF /* インポズ入力OFFで検知

ALM_ON /* アラーム入力ONで検知

ALM_OFF /* アラーム入力OFFで検知

無効にするには値を設定しません。

INSET_314 X_A ALM_ON¦INP_ON¦LMT_OFF /*ｱﾗｰﾑ,ｲﾝﾎﾟｽﾞ有効

INSET_314 X_A LMT_OFF /*ｱﾗｰﾑ,ｲﾝﾎﾟｽﾞ無効

マルチタスクでパルス発生

タスク ごと に独立した軸制御するこ と ができます。次に例と してタスク 1でX軸、タスク 2でY軸を制御します。

この例のよ うにMPG-314は軸ごとに異なるタスクからの制御も可能です。このため、MPG-314では搭載され

た4軸すべてを無駄なく使い切るこ とができます。

10 FORK 1 *YAXIS
20 PG &H410
40 ACCEL X_A -25 4000 400 100/* ↓ X軸の制御
50 RANGE X_A 50000 -50000
60 INSET_314 X_A SLMT_ON|LMT_ON
70 STPS X_A 0
80 DO
90 RMVS X_A 1000
100 WAIT RR(X_A)==0
110 RMVS X_A -1000
120 WAIT RR(X_A)==0
130 LOOP
140 *YAXIS
150 PG &H410
160 ACCEL Y_A -25 8000 400 100/* ↓ Y軸の制御
170 RANGE Y_A 50000 -50000
180 INSET_314 Y_A SLMT_ON|LMT_ON
190 STPS Y_A 0
200 DO
210 RMVS Y_A 1000
220 WAIT RR(Y_A)==0
230 RMVS Y_A -1000
240 WAIT RR(Y_A)==0
250 LOOP

INSET_314( 条件設定 )について

INSET_314はリ ミ ッ ト 、アラーム、ソフト リ ミ ッ ト 、インポジショ ン入力の動作を設定します。このう ちリ ミ ッ

ト については、 無効の設定がせできないので注意してく ださい。 設定できるのはON/OFFの論理だけです。

INSET_314コマンドはMPG-314内部のMCX-314へのレジスタ(16bit)設定コマンドです。 このため指定するパ

ラ メータはビッ ト変数となっています。複数設定する場合は、"¦"演算子で結合します。また、 コマンドがレ

ジスタ設定であるために常に上書きです。

10 INSET_314 X_A INP_ON|SLMT_ON
20 INSET_314 X_A ALM_ON

前記のよ うに実行する と最終的には20のALM_ONのみが有効とな り INP_ON,SLMT_ONも無効になり ます。

10 INSET_314 X_A INP_ON|SLMT_ON| ALM_ON

前記のよ うにすればインポズ、 ソフ ト リ ミ ッ ト 、 アラームと も有効にな り ます。

動作範囲設定 (ソフトリミット )

ソフ ト リ ミ ッ トの有効無効をINSET_314の再実行で設定するのはかえって面倒です。SLMT_ONはあらかじめ

設定しておいてRANGEコマンドでその都度範囲を設定して使用するほうが扱いが簡単になり ます。

10 RANGE X_A +800000,-800000 /* 実働より広い範囲
20 INSET_314 X_A INP_ON|SLMT_ON| ALM_ON

とすれば事実上、 レンジは規制されないのと同じこ とにな り ます。 プログラム進行中に、 必要に応じて、

100 RANGE X_A +1000,-2000

と設定すればその時点から リ ミ ッ トが有効に動作します。

アラーム、リミット入力の読み取り

プログラム動作中に各入力が検出されたかど うかは次のよ うにRR()関数をもちいます。たとえば、ソフ ト リ

ミ ッ トが動作した状態では、RR(X_A,SLMT@)の値が0でなくな り ます。読み取り ビッ ト定数には末尾に'@'の

ついたものと'_'ものがあ り意味が異なり ます。 @のついたものは状態監視読み取り、つま り、 その入力ポー

トが現在どのよ うな状態になっているかを返すものです。 ALM_ONと設定されていて入力ポート をONにす

る とRR(X_A,ALM@)は0でない値を返します。 これに対して、 ALM_等の_のついたビッ ト定数は、パルス発生

が停止した原因を知らせるためのものです。 たとえば、 ALM_ON を設定し、 入力を実際に ON する と

RR(X_A,ALM@)は0でない値を返しますが、RR(X_A,ALM_)は実際にパルス発生をして停止させるまでは0のま

まです。読み取りは、 ビッ ト定数をORするこ と もできます。 この場合は、いずれかの入力を検出する と0で

な くな り ます。

RR(X_A,ALM@|SLMP@|SLMM@)

しかし、 ALM@¦LMTP_¦LMTM_とい う よ う な種類の異なるビッ ト定数を混在させるこ とはできません。

INCHK_314 について

配線直後に入力ポートの様子を確認するコマンドです。 実行する と次のよ うに表示されます。

X_S1～U-INPまでは、実際の入力論理値を表示しますが、XLMT+～ULMT-では、設定した論理値にしたがっ

て リ ミ ッ ト が動作している場合にONと表示します。

パワーオン後、 何も設定しないでこのコマンドで LMT を確認する と、 ON で有効というデフォルト状態に

なっているために、 LMTポート をONする とONと表示します。

#INCHK_314
X_S1 __ Y_S1 __
Z_S1 __ U_S1 __

ビット定数 役割 パラメータを設定しない場合 関係するビット定数

[インポジション]
INP_ON
INP_OFF

サーボのインポジション入力を有効に
て論理設定します

インポジション無効

[アラーム]
ALM_ON
ALM_OFF

サーボからのアラーム力を有効にて論
理設定します

アラーム無効 ALM@
ALM_

[ソフトリミット]
SLMT_ON

ソフトリミットを有効にします。無効
の場合は引数を設定しません

ソフトリミット無効 SLMP@
SLMM@
(注: Pは+,Mは-)

[(ハード)リミット]
LMT_ON
LMT_OFF

ハードリミットの論理を設定します。
デフォルトではONでリミットとなりパ
ルス発生しなくなります。

ONでリミット有効 LMTP@
LMTM@
LMTP_
LMTM_

[パルス出力]
MD_2PLS
MD_DPLD

デフォルトではMD_2PLS方式、つま
りCW/CCW方式です。方向指示パルス入
力の時はMD_DPLSを引数に追加します。

CW/CCW方式

X_S2 __ Y_S2 __
Z_S2 __ U_S2 __
XIN2 __ YIN2 __
ZIN2 __ UIN2 __
XIN3 __ YIN3 __
ZIN3 __ UIN3 __
X-INP __ Y-INP __
Z-INP __ U-INP __
X-ALM __ Y-ALM __
Z-ALM __ U-ALM __
XLMT+ __ XLMT- __
YLMT+ __ YLMT- __
ZLMT+ __ ZLMT- __
ULMT+ __ ULMT- __

多軸同時制御と直線補間

■RMVS,MOVSコマンド

4つのパラ メータを同時に指定してパルス発生するこ とができます。動作させない軸には0をセッ ト します。

RMVS 1000 0 2000 0 /*XU軸のﾊﾟﾙｽ発生をします。
MOVS 1000 2000 VOID VOID /*絶対位置管理のﾊﾟﾙｽ発生の場合は動作させない軸に

/*対しては、VOIDを設定する。
/*(0では0位置を意味するため)

■RMVL,MOVLコマンド

直線補間の場合は、RMVL,MOVLを用います。 この場合はMPG-314の機能上三軸までとな り ます。四軸直線

補間はできません。

RMVL 1000 2000 3000 0 /* XYU直線補間　X軸の速度に従います。
MOVL VOID 1000 2000 3000 /* YUZの直線補間　Y軸の速度に従います。

■原点復帰

原点復帰にはJ4のIN0,IN1を使用します。 IN0をニアオ リ ジンと しIN1をサーボのZ(C)相に接続します。

IN1側はJ2(23～26)に設けられた、別端子を使用するこ とによ り差動入力対応とするこ とができるためです。

HOMEには次のよ うにHOMEコマンドを使用します。条件設定はビッ ト設定を用います(IN0_ON/OFF～IN3_

ON/OFF)。

条件はOR条件です。 またHOMEで設定された停止条件は残ってしまいますので、STOPコマンドでク リ アす

る必要があ り ます。

HOME X_A -1000 IN0_ON|IN1_OFF /* IN0がONもしくはIN1_OFFになったら停止

通常、 ニアオ リジン、 オ リ ジンとい う順にな り ますので次のよ うにな り ます。

HOME X_A -1000 IN0_ON /* スリット検出
WAIT RR(X_A)==0
HOME X_A -100 IN1_ON /* Z相検出
WAIT RR(X_A)==0
STOP X_A VOID /*停止条件クリア
STPS X_A 0 /*現在位置クリア

また、 HOMEコマンドは二軸まで対応しますので次のよ うにも記述できます。

10 HOME X_A -1000 IN0_ON Y_A -1000 IN0_ON /* ｽﾘｯﾄ検出
20 WAIT RR(X_A|Y_A)==0
30 HOME X_A -100 IN1_ON Y_A -100 IN1_ON /* Z相検出
40 WAIT RR(X_A|Y_A)==0
50 STOP X_A|Y_A VOID /*停止条件クリア
60 STPS X_A 0 : STPS Y_A 0 /*現在位置クリア

ニアオリジンまでの高速原点復帰は次のよ うな記述ができます。

10 STOP X_A IN0_ON /*停止条件＝スリット検出
20 RMVS X_A 10000 /*10000パルス発生までに上記条件が成立したら減速停止
30 WAIT RR(X_A)==0 /*停止待ち
40 STOP X_A VOID /*条件クリア

溜まりパルスのリセット

原点復帰速度が十分に低速で、 機械系のバッ ク ラ ッシュなどが少なければ前記のよ う な原点復帰で十分な

のです。しかし、溜ま りパルス(サーボ追従の遅れ)がある場合、原点復帰の成立と同時に溜ま りパルス リセッ

ト信号を出力する必要があ り ます。

10 '原点復帰で残パルスクリアの例
20 'Near Orgin
30 HOUT &HF0 : ' Servo On Servo reset OFF
40 HOME X_A 100 IN0_ON
50 HOME Y_A 100 IN0_ON
60 HOME Z_A 100 IN0_ON
70 HOME U_A 100 IN0_ON
80 WAIT RR(ALL_A)==0
90 'Detect Z phase
100 '必要に応じて他タスクpause
110 ' pause 3 4 5
120 HOME X_A 10 IN1_ON
130 HOME Y_A 10 IN1_ON
140 HOME Z_A 10 IN1_ON
150 HOME U_A 10 IN1_ON
160 a=0
170 'RR(ALL_A)の値をHOUTに反映
180 'Zのビットが4Uが8となるので注意
190 '論理反転してリセット信号とする。
200 WHILE a<>&HFF : HOUT a=(15-RR(ALL_A))|&HF0 : WEND
210 TIME 10
220 HOUT &HF0 : 'カウンタリセットクリア
230 ' cont 3 4 5 

ここではOP1～OP4をサーボリ セッ ト 信号(X,Y,Z,Uの順)と して使用します。 MPG-314のビジービッ ト の値を反

転して、OP1～OP4に複写すると いう 方法です。速度を早めるために一行の式で記述しています(ステップ200)。

また、 動作している他のタスクは応答速度を悪化させますので、 出来る限り PAUSEなどで停止させて下さい。

カウンタ機能

MPG-314には4軸のロータ リーエンコーダ用カウンタ機能が付属しており、OC5～OC8にフォ ト カプラを装

着するこ とによ り使用可能とな り ます。

カウンタに関係するコマンド と関数は次のとおりです。

◆STPS AXIS Val 

AXIS ： カウンタ指定 X_C～U_C

Val ： 設定したい値。

注意: AXISにX_A～U_Aを用いる と現在位置の意味になり ます。

◆X(-1)～U(-1) 

座標関数に-1を設定する とカウンタ値を返します。

◆X(-2)～U(-2) 

座標関数に-2を設定する とカウンタ値を返し、 現在のカウンタをク リ アします。

X(-2,Fac)～U(-2,Fac) のよ うに引数を追加する と"カウンタ値×Fac"を返します。 -1の場合も有効です。

次のプログラムは、エンコーダをJOGダイヤルに使用した例です。Aに倍率を入れておきます。Aの値を多く

すれば移動量が多くな り ます。 このよ うに関数と コマンドを一行にま とめてしま う と高速になり ます。

480のスワ ップは、 カウンタが変化しない場合、無限ループ実行にな り全体の実行効率が低下するのを回避

するためです。 なおカウンタの値をま とめてク リ アする場合は、

STPS X_C|Y_C 0 /* XY同時クリア
STPS ALL_C 0 /* 全クリア

とい う記述ができます。

--初期化--
10 PG &H410
20 PG &H410 1
30 A=1
40 INSET_314 X_A|Y_A 0 
----------
330 ACCEL 8000
340 STPS X_C 0
350 STPS Y_C 0
360 FORK 1 *jog2
370 END

440 *jog2
450 WHILE A<>0
460 RMVS X(-2,A) Y(-2,A) 0 0
470 WAIT RR(X_A|Y_A)==0
480 SWAP
490 WEND
500 PRINT "Jog End"
# 

注(1):

10～40は初期値と して必要な設定です。 INSET_314はカウンタをA-B2相方式に設定します。 この例ではダ

ミーと して0を引数と していますが、パラ メータはどのよ うな値でもかまいません。INSET_314が実行されて

いないと2相信号をカウン ト しません。

注(2):

カウンタはデフォルトで"4てい倍"となっています。 (2倍,1倍)率を与えるにはINSET_314の設定値に&H400(2

倍)、 &H800(1倍)をORします。

INSET_314 X_A &H800|INP_OFF /*1倍で読み取り

条件停止

STOPコマンドには二通りの使用方法があ り ます。

ひとつはインタプ リ タ上で、 適宜 STOP コマンドを実行しプログラム制御でパルス発生を停止する場合で

す。 も うひとつは、 IN0～IN3のハードで用意された自動停止を使用する場合です。

■ソフト停止 

◆STOP AXIS CMND

AXIS : X_A ～ U_Aを指定します。

CMND : STP_D(減速停止),STP_I(急停止)のいずれかを設定します。

10 PG &H410
20 ACCEL X_A 4000
30 CLRPOS
40 MOVS X_A 10000
45 WHILE RR(X_A)<>0
50 IF SW(1)==1 THEN : STOP X_A STP_D : END_IF
60 WEND 

前記の例は動作中にセンサ入力を検出したら減速停止させる という ものです。

■ハード入力検出停止

◆STOP AXIS cond

インタプ リ タで制御する停止は自由度が高いのですが、応答速度に限界があ り、正確な検出停止には使えま

せん。 MPG-314のJ2,J4に用意されているIN1～IN3の入力を用いる とハード的に入力を検出して停止するた

め、 正確さ高速さを要求される停止機能に対応します。

40 ACCEL X_A 4000
50 CLRPOS /* 現在値クリア
60 STOP X_A IN3_ON /* STOPの条件設定
70 MOVS X_A 10000 /* パルス発生
80 WAIT RR(X_A)==0 /* 停止待ち
90 STOP X_A VOID /* STOPの設定を解除

100 PRINT X(0) /* 現在値表示
110 END 

この例はJ2側のIN3がONになったらパルス発生を停止します。 停止したら必ずSTOPの設定を解除します。

条件はOR演算子"¦"で結合できますが、 AND条件はICの機能に無いためサポート されていません。

予約定数

■指定ｸﾞﾙｰﾌﾟ

■状態監視ｸﾞﾙｰﾌﾟ ■停止条件ｸﾞﾙｰﾌﾟ

NOP ﾉｰｵﾍﾟﾚｰｼｮﾝ。軸選択のﾚｼﾞｽﾀ設定に使用。

VOID 入力無効定数

CLR_ERR ｺﾏﾝﾄﾞｴﾗｰ解除

STP_I
STP_D 

停止ｺﾏﾝﾄﾞ
   STP_I : 急停止    STP_D : 減速停止

X_A,Y_A,
U_A,Z_A,
ALL_A

軸指定定数
   X_A : X軸        Y_A : Y軸         U_A : U軸
   Z_A : Z軸        ALL_A : 全軸 (ACCEL等で)

X_C,Y_C,
U_C,Z_C

カウンタ指定定数
   X_C : X軸        Y_C : Y軸
   U_C : U軸        Z_C : Z軸

CW
CCW

円弧補間方向指定。連続補間に使用。
   CW : 時計方向      CW : 反時計方向

DS_DACL
EN_DACL 

自動減速無効、有効。連続補間に使用。
   DS_DACL : 無効    EN_DACL : 有効

IN0_ON,IN1_ON,IN2_ON,IN3_ON,
IN0_OFF,IN1_OFF,IN2_OFF,IN3_OFF

停止入力をON/OFF状態で有効にする。
   HOME X_A,STOP X_Aﾀｲﾌﾟのｺﾏﾝﾄﾞで使用。

INP_ON
INP_OFF 

ｲﾝﾎﾟｼﾞｼｮﾝを有効にしてON/OFFを決定する。
   INP_ON : lowでｱｸﾃｨﾌﾞ    INP_OFF : Hiでｱｸﾃｨﾌﾞ

ALM_ON
ALM_OFF

ｱﾗｰﾑを有効にしてON/OFFを指定
   ALM_ON : lowでｱｸﾃｨﾌﾞ    ALM_OFF : Hiでｱｸﾃｨﾌﾞ

LMT_ON,MT_OFF ﾘﾐｯﾄONの時に成立

SLMT_ON RANGEｺﾏﾝﾄﾞによるｿﾌﾄﾘﾐｯﾄの有効化定数

MD_2PLS CW/CCWﾊﾟﾙｽ発生指定

MD_DPLS DIR ﾊﾟﾙｽ発生指定

EMG_ EMGによって停止

ALM_ ｱﾗｰﾑ入力によって停止

LMTP_ ﾘﾐｯﾄ入力(+)によって停止

LMTM_ ﾘﾐｯﾄ入力(-)によって停止

IN3_ STOP条件のIN3によって停止

IN2_ STOP条件のIN2によって停止

IN1_ STOP条件のIN1によって停止

IN0_ STOP条件のIN0によって停止

EMG@ EMGがｲﾈｰﾌﾞﾙ検出

ALM@ ｻｰﾎﾞｱﾗｰﾑ出力ON検出

LMTP@ ﾘﾐｯﾄ入力(+)検出

LMTM@ ﾘﾐｯﾄ入力(-)検出

SLMP@ ｿﾌﾄﾘﾐｯﾄ(+)検出

SLMM@ ｿﾌﾄﾘﾐｯﾄ(-)検出

レジスタ概要

■読み取り用

◆RR0:パルス、 エラ－監視

パルス発生中の監視、エラーの監視用です。関数RR(0)で参照するこ とができます。下位4ビッ トがビジー、次
の4ビッ トがエラーです。 次のよ うに使用します。 エラーは次のRR1の上位8ビッ トの反映です。
パルス発生の終了待ち

WAIT RR(0)&X_A=0
も し くは 

WAIT RR(X_A)=0

他のタスクのRR0の読み取り

RR(&H410)

◆RR1:軸毎のエラーステータス1

X,Y,U,Zそれぞれか持つステータスです。関数RR(軸定数,1)とい う形式で読み取るこ とができます。このレジ
スタの上位8ビッ トはパルス発生がどのよ うに終了したかを表します。 RR()関数で読み取り ますが次のよ う
な使用方法とな り ます。
X軸リ ミ ッ ト エラーの参照

LMT=&h3000&RR(X_A,1)

◆RR2:軸毎のエラーステータス2

X,Y,U,Zそれぞれが持つエラー入力の参照レジスタです。 ALAMやLMT信号を直接参照するにはこのレジス
タを参照します。 

◆RR3:軸毎のエラーステータス3

X,Y,U,Zそれぞれが持つカウンタの比較情報です。MCX-314には軸ごとに比較用レジスタ と比較器があ り ま
す。 これによ り特定の場所で正確な信号を発生させるこ と も可能です。

■設定用

◆WR0:コマンドレジスタ

軸選択とMCX-314に対するコマンドレジスタです。例えばHOUT X_A;STP_Dとい う記述では、X_A;STD_Dと
い う演算によって&H0126 とい う数値が生成されますがHOUTはこれをWR0への直接書き込みと して扱いま
す。WR0は軸選択レジスタ と しても使用されるため、次のWR1,WR2への書き込み時にもダ ミーライ トが必要
とな り ます。

◆WR1:モードレジスタ1

主に IN0 ～ IN3 のド ラ イブ停止検出信号の設定と して使用します。 WR1 への書き込みは次のよ う なフォー
マッ トにな り ます。

HOUT 2 (X_A;NOP,1)

2はWR2に書き込む値でIN0がローで停止という条件です。 ()のの中の二番目の引き数X_A;NOPでWR0に対し
て軸設定をします。','の後ろの番号がレジスタ番号です。レジスタ番号にボードアドレスを加える と任意の
アドレスのMPG-314にアクセスするこ とができます。

◆WR2:モードレジスタ2

リ ミ ッ ト検出やインポジシ ョ ン検出の設定を行う レジスタです。 しかしながら このレジスタにはパルス出
力モードの設定ビッ ト も含まれていますので注意して再設定して ください。

 HOUT INP_ON+MD_2PLS (X_A;NOP,2)

こ こではINP_ON+MD_2PLSをWR2に設定する値と しています。 MD_2PLSはパルス出力モードをCW/CCW負
論理方式(デフォルト )に設定するビッ ト定数ですが、も し この値を加えておかないとパルス出力のモード設
定フ ィルードがすべて0とな り、パルス出力が正論理に変わってしまいます。(X_A;NOP,2)はWR1の場合と同
様の入力でX軸選択でWR2を選んでいます。

◆WR3:モードレジスタ3

下位3ビッ トが加減速度のモード設定とな り ます。初期値と してこのレジスタは4(自動減速、対称減速、S字
加減速度有効)が設定されています。 このほかにはOUT4～7の出力モード設定があ り汎用出力と して使用す
るか、 位置信号の出力と して使うかを決定できます。
WR3のビッ ト配置

0000 ¦ 7 6 5 4 ¦ 0 X X E1 ¦ E0 S D1 D0 ¦

上位4ﾋﾞｯﾄ:無効

7～4 ： 出力ﾎﾟｰﾄ(J2の17～22)。 下位2ﾋﾞｯﾄのみ使用

E1,E0:MPG-314では非接続なので0とする。

S:S字有効=1

D1:減速レジスタの選択。 通常0

D0:自動加減速の有効 ・無効。 1で無効。

S 字加減速

S字加減速を使う と滑らかな始動 ・停止が出来て、 速く インポジシ ョ ンを得られる場合があ り ます。 しかし
設定を大き くする と加減速領域が広がり、かえってタク ト タイムが長くなるこ と もあ り ます。適切にド ライ
バのゲイン調整を行い装置剛性に合ったS字領域を設定しなければな り ません。 次はMPG-314でS字加減速

をした時の実測例です。

ｻｰﾎﾞﾓｰﾀ 安川 SGM-A3BW12C

ｻｰﾎﾞﾊﾟｯｸ 安川 SGDA-A3BP

ﾓｰﾀは無負荷

ﾓﾆﾀﾘﾝｸﾞｿﾌﾄ Sigma Win Ver1.20

ﾌﾟﾛｸﾞﾗﾑ

10  ACCEL X_A -1 100000  /*第2ﾊﾟﾗﾒｰﾀを変える
20  DO
30    RMVS X_A 40000
40    WAIT RR(X_A)==0
50    TIME 50
60  LOOP

 

 

ACCEL X_A -25 100000
S字区間25%

ACCEL X_A -1 100000
S字区間無し

ACCEL X_A -50 100000
S字区間50%

引きずり現象について

特定の設定値(ACCEL)の特定の速度(FEED)でS字加減速度が有効になっている と、減速が異常に早く始ま り

最低速度でひきずるよ うに動作する場合があ り ます。

ACCEL X_A -10 250000 20000 2000 : FEED X_A 220 での実測

FEED設定が大きい(スピードが遅い)と発生する傾向があ り ます。台形駆動ではこの現象はあ り ません。詳細

は巻末の付録参照。

応用例

■一枚のMPG-314の軸をタスクごとに非同期で使用する場合

MPG-68K/405では1枚のMPGに対して複数のタスクから同時にパルスコマンドを実行できませんがMPG-

314では可能です。 次のプログラムは1枚のMPGの各軸を別々のタスクで制御しています。

MPG-68K/405でこのよ うな制御を行おう とする と軸数分のMPGが必要だったので省スペース、 低コス トに

な り ます。

  PG &H410 1 /*複数のタスクを一枚のMPGに割り当てます。
  PG &H410 2
  PG &H410 3
  PG &H410 4
  FORK 1 *MPG_x /*それぞれのタスクを起動します。
  FORK 2 *MPG_y
  FORK 3 *MPG_u
  FORK 4 *MPG_z
  END 
*MPG_x
  ACCEL X_A 8000 /*加速度を軸毎に選択します。
  STPS X_A 0
  DO 
    GOSUB *PULSE X_A /*タスク間共通サブルーチンをコールしていますが、
  LOOP /*例文を簡略化するためです。
*MPG_y
  ACCEL Y_A 6000
  STPS Y_A 0
  DO 
    GOSUB *PULSE Y_A
  LOOP 
*MPG_u
  ACCEL U_A 4000
  STPS U_A 0
  DO 
    GOSUB *PULSE U_A
  LOOP 
*MPG_z
  ACCEL Z_A 2000
  STPS Z_A 0

 
引きずり

  DO 
    GOSUB *PULSE Z_A
  LOOP 
*PULSE /*タスク間で共通のルーチンです。
  _VAR axs! /*そのためにローカル変数を用いています。
  FOR i!=0 TO 10 /*単軸制御は例のように軸選択定数を用いて制御するのが安全です。
    FEED axs! i!*20
    RMVS axs! 1000
  NEXT i!
  WAIT RR(axs!)==0
  MOVS axs! 0
  RETURN

■一枚のMPG-314の軸で三軸直線補間と残りの軸を他のタスクで制御

次のよ うに直線補間と単軸駆動を組み合わせて非同期で使用できますが、XY,ZUよ うに二軸直線補間を二組

同時に使用するこ とはできません。

  PG &H410 1 /*二つのタスクを一つのMPGに割り当てます
  PG &H410 2
  FORK 1 *MPG_xyz
  FORK 4 *MPG_z
  END 
*MPG_xyz
  ACCEL X_A 8000 /*主軸(X>Y>Z>U)のACCELを設定
  STPS 0 0 0 VOID /*他のタスクが使う軸はVOID変数で扱わないようにする。
  DO 
    FOR i=1 TO 10
      RMVL i*1000 i*100 i*-10 0 /* RMVL,MOVLは直線補間
    NEXT I
    WAIT RR(X_A)==0
    MOVL 0 0 0 VOID
  LOOP 
*MPG_z /* Z軸の単独制御
  ACCEL Z_A 2000
  STPS Z_A 0
  DO 
    GOSUB *PULSE Z_A
  LOOP 
*PULSE
  _VAR axs!
  FOR i!=0 TO 10
    FEED axs! i!*20
    RMVS axs! 1000
  NEXT i!
  WAIT RR(axs!)==0
  FEED axs! 0
  MOVS axs! 0
  RETURN

■パルス発生  途中停止や速度変更

入力を検出してソフ トで停止させる

MPG-314 ネイティブコマンドは実行後すぐに次のステップに移行します。 次はパルス発生しながらセンサ

がオンするのを待ちます。 このよ うにパルス発生とその後の管理が同一タスクで行えるのでプログラムの

見通しがよ くな り ます。

*Y
RMVS Y_A 200000 /*相対座標移動
WAIT HPT(1)=1 /*センサ入力待ち
STOP Y_A STP_I /*STP_Iで即停止、STP_Dで減速停止です。
WAIT HPT(1)=0
MOVS Y_A 0
WAIT RR(0)&Y_A=0
TIME 500
GOTO *Y

条件によ り速度を変更する

行き先の決まっているパルス発生でS字加減速を設定している と減速のみで増速できません。RMVCではS字

加減速でも増速 ・減速いずれも可です。

PG &H410
ACCEL 8000
CLRPOS
RMVS X_A 3000
WAIT X(0)>1000
FEED X_A 128 /*速度を途中で変更する。
WAIT X(0)>2000
FEED X_A 0

原点復帰

原点復帰はHOMEコマンドを使用する と明示的になり ますが、現在位置については放置されます。原点復帰

が完了したら位置をク リ ア、 あるいは設定する必要があ り ます。

PG &H410
ACCEL 8000
HOME X_A IN0_ON -125 Z_A IN0_ON -200 /*速度指定原点復帰 IN0使用 FEED 125 CCW FEED 
WAIT RR(X_A|Z_A)==0 　200 CCW
HOME X_A -1000 IN1_ON Z_A -500 IN1_ON /*定則原点復帰 IN1使用
WAIT RR(X_A|Z_A)==0
PRSET_ACCEL X_A Z_A /* ACCELパラメータ復帰 停止条件クリア
FEED X_A 0 Z_A 0 /* HOMEで変更したFEEDの変更
STPS 1000 VOID VOID 1000 /*位置設定

■パルス発生  途中停止  ハード

MCX-314には入力ポート IN0-IN3を使用した停止条件を決定するレジスタWR1があ り ます。 このレジスタに

適切な値を設定するこ とによ り、 各軸ごとに停止条件を定めるこ とができます。 IN0,IN1はS1,S2に割り当て

られ、IN2は差動入力ポート、 IN3は拡張入力ポートに割り当てられています。 これによ り必要に応じてハー

ド停止させる こ とができるためホス ト側の遅れ時間を気にするこ とな く確実にパルス発生を停止させるこ

とができます。 このWR1を設定するためにSTOP拡張コマンド と IN0_ON～IN3～OFF定数が用意されていま

す。

 PG &H410
 ACCEL 10000
 STPS X_A 0
 STOP X_A IN3_ON|IN1_OFF /* IN1がOFFになるかIN3がオンになると停止
 RMVS X_A 1000000 
 STOP X_A VOID /*停止条件の解除

INPOSの有効 サーボオンエラーリセッ ト など

INPOSを有効にする とｻｰﾎﾞのINPOS信号と連動させるこ とができます。 INPOSの設定はWR2レジスタで行い

ます。レジスタの意味は次のとおりです。なおこのレジスタを操作するのに必要な定数セッ トが用意されて

います。

X軸のインポジシ ョ ンをONで有効、 ALMをONで有効、 LMTをONで検出とするには

wr2=MD_2PLS|INP_ON|ALM_ON|LMT_ON
hout wr2 (X_A;NOP,2)

MD_2PLS

MD_DPLS

デフォル トのCW/CCW負論理パルス方式です。

DIR/PULSE方式です。

INP_ON

INP_OFF

インポジ ョ ンをON入力待ちと します。

インポジ ョ ンをOFF入力待ちと します。

ALM_ON

ALM_OFF

アラームをONで発生と します。

アラームをOFFで発生と します。

LMT_ON

LMT_OFF

リ ミ ッ ト信号をONで発生と します。

リ ミ ッ ト信号をOFFで発生と します。

0       1000     2000    3000

減速　増速

wr2を設定する場合はMD_2PLSをかならず加えて設定しなおして ください。 また、 パルスDIR/PULSE方式に

するにはMD_2PLSの替わりにMD_DPLSを使います。

Z軸のINPOSを有効とするには

10 PG &H410
20 ACCEL 10000
30 CLRPOS
35 HOUT MD_2PLS|INP_ON  (Z_A;NOP,2)
40 DO
50 RMVS Z_A 10000
60 RMVS Z_A -10000
70 LOOP 

■移動距離を規定しないパルス発生 (RMVC,PLSC)

原点復帰やテープの定速送りでパルス量を規定しないパルス発生があ り ます。 この場合はRMVC,PLSCを用

います。RMVCはACCELの指定に従います。減速停止と組み合わせて次のよ うに使用できます。また原点復

帰の場合座標値のク リ アが必要ですが、 STPS X_A 0、 STPS VOID 0 0 VOIDなどのよ うに実施します。

PG &H410
ACCEL 10000
FEED X_A 64 : FEED Z_A 32
RMVC Z_A 1 X_A -1 /* ZをCW,XをCCWで回転。ACCELに従って加速、FEEDに従った速度
TIME 1000
STOP Z_A STP_D /* Zを減速停止
TIME 1000
STOP X_A STP_I /* Xを急停止

次はIN0～IN3を用いた例です。

PG &H410
ACCEL 10000
FEED X_A 64 : FEED Z_A 32
STOP X_A IN0_ON /* X軸停止設定XS1
STOP Z_A IN0_ON /* Z軸停止設定ZS1
RMVC Z_A 1 X_A -1 /* XS1(IN0)検出減速停止
WAIT RR(X_A|Z_A)==0 /*停止待ち
STOP X_A VOID /*停止設定解除
STOP Z_A VOID /*停止設定解除

RMVCの替わりに PLSCを用いる と加減速なしの定速パルス発生になり、 ppsを直接指定できます。

しかし、PLSCコマンドはACCELの設定を壊してしま うため、実行後RSTRコマンドでACCEL設定値を戻す必

要があ り ます。 RSTRには設定された停止条件を解除する機能も含まれています。

PG &H410
ACCEL 10000
STOP X_A IN0_ON
STOP Z_A IN0_ON
PLSC Z_A 100 X_A -100 /*100pps指定
WAIT RR(X_A|Z_A)==0 /*停止待ち
PRSET_ACCEL X_A Z_A /*パラメータ復帰。また停止条件解除

■エラーの扱い方

ERR_PAUSEを使用する と発生したエラーに対応してタスクを自動停止させるこ とができます。 PG出力でエ

ラーとなった場合、パルス発生そのものはされませんが、 インタプ リ タは動作しつづけます。 このため、ア

プリ ケーシ ョ ンによっては装置を誤作動させてしま う場合があ り ます。 ERR_PAUSE をセッ ト しておけば確

実にインタプ リ タが停止しますので安全な対応をと るこ とができます。

10 PG &H410 1
20 FORK 1 *prg
30 DO
40 IF TASK(1)==-3 THEN /*停止を検出

50 PRINT "Err happen!!" : TIME 1000
60 HOUT X_A;CLR_ERR &H410 /*エラーのクリア
70 CONT 1 /*（実際にはもう少し複雑な手続きの上、
80 END_IF /*  再起動か一時停止の解除とする）
90 LOOP
100 END
110 *prg
120 ACCEL 8000
130 wr2=MD_2PLS|ALM_ON /*ALAM信号を有効にします。
140 HOUT wr2 (X_A;NOP,2) /*X軸のWR2を書き換えます。
150 ERR_PAUSE X_A /*X軸のエラー検出をイネーブルします。
160 DO
170 RMVS X_A 1000 /*実行プログラム
180 WAIT RR(X_A)==0
190 RMVS X_A -1000
200 WAIT RR(X_A)==0
210 LOOP

RS-232のプログラム

MPC 同士での通信例

最近のシステムはますますインテ リ ジェン ト化され画像処理装置、表示装置、パソコン、 Ｌ ＡＮなどの外部

機器と通信で結ばれ多くのデータの授受が行われています。MPC-684にはRS-232ポートが3チャンネル装備

されておりそのうち2チャンネルをユーザーが使用できます。 PRINT文やINPUT文による文字列単位の送受

信、 INP$ ・ PUT#によるキャラ ク ター単位の送受信、 さ らにLEN ・ STR ・ STRCPY ・ VALなどの関数 ・ コマン

ドによる柔軟な文字列処理で複雑な通信フォーマッ トにも対応できます。次のプログラムはMPC-684 2台の

CH0同士を接続し、 ＭＰ Ｃ－Ａから送信されるデータをＭＰ Ｃ－Ｂで解析しています。

■ＭＰＣ－Ａ（送信側）

送信フォーマッ ト

［SX］ ［変数名］ ［*］ ［データ］ ［*］ ［チェ ッ クサム］ ［EX］ ［ET］

概 要

最初にEQコードを送信しAKコードの応答を待つ （受信側のREADY確認）。その後フォーマッ トに従

い変数名 ・データ ・ チェッ クサムを送信する

  CNFG#0 "9600b8pns1XON"
'
  PRINT#0 CHR$(&H5) ’EQコード送信
  WAIT INP$#0(1)==CHR$(&H6) ’AKコード受信待ち
    ABCDE=12356
    FGHIJ=98765
    A$="FGHIJ" ’A$に文字列FGHIJを入れる
    B$=STR$(FGHIJ) ’B$にFGHIJの値を文字列として格納
  PRINT#0 CHR$(&H2) A$ "*" B$ "*" ’SXコードに続きデータを出力
    CS=0
  FOR I=0 TO LEN(A$) ’A$のチェックサムを計算
    STRCPY A$ C$ I 1
     CS=CS+ASC(C$)
  NEXT I
  FOR I=0 TO LEN(B$) ’B$のチェックサムを計算
    STRCPY B$ C$ I 1
      CS=CS+ASC(C$)
  NEXT I
    CS=CS+ASC("*")
    CS$=HEX$(CS&HFF) ’チェックサムの値を文字列としてC$に格納
  PRINT#0 CS$ CHR$(&H03) ’C$とEXコードを出力
  PRINT#0 CHR$(&H4) ’ETコードを出力

■ＭＰＣ－Ｂ（受信側）

受信フォーマッ ト

［SX］ ［変数名］ ［*］ ［データ］ ［*］ ［チェ ッ クサム］ ［EX］

概　要

EQコードを受信したらAKコードを返してREADY状態を知らせ受信を開始する。 EXコード受信によ

りチェ ッ クサムデータを比較、 ETコードで終了する。

  CNFG#0 "9600b8pns1XON"
'
  WAIT INP$#0(1)==CHR$(&H5) ’EQコード待ち
  PRINT#0 CHR$(&H6) ’AKコード出力
     A$=""
*LOOP
  WAIT LOF(0)<>0
    A$=INP$#0(1)
  SELECT_CASE A$ ’SXコードなら受信ETコードなら終了
        CASE CHR$(&H2) : GOTO *CASE1

        CASE CHR$(&H4) : GOTO *CASE2
  END_SELECT
  GOTO *LOOP
*CASE1
    C=0
    A$=""
    B$=""
    C$=""
  DO UNTIL A$=="*" ’変数名を受信
      B$=B$+A$
      A$=INP$#0(1)
      C=C+ASC(A$) ’チェックサムデータ加算
  LOOP
  PRINT B$
    A$=""
  DO UNTIL A$=="*" ’データを受信
      C$=C$+A$
      C=C+ASC(A$)
      A$=INP$#0(1)
  LOOP
  PRINT C$
    C=C&&HFF
    D$=HEX$(C) ’チェックサムデータを文字列に変換
    A$=INP$#0(2) ’チェックサムデータを受信
  PRINT D$ A$  
  WAIT INP$#0(1)==CHR$(&H03) ’EXコード待ち
  IF D$<>A$ THEN
    PRINT ”チェックサムエラー”
    PRINT#0 CHR$(&H15) ’チェックサムエラーならNKコード出力
    GOTO *LOOP
  END_IF
  SELECT_CASE B$ ’変数にデータを格納
    CASE "ABCDE"
      ABCDE=VAL(C$)
    CASE "FGHIJ"
      FGHIJ=VAL(C$)
    CASE_ELSE
    PRINT ”カイドクデキマセン”
  END_SELECT
  GOTO *LOOP
*CASE2
  PRINT "ABCDE=" ABCDE
  PRINT "FGHIJ=" FGHIJ

RS-232 のエラー処理

RS-232 で通信を行う場合にはデータの信頼性が重要になり ます。 どんなに高度な周辺機器と接続されても

データがデタラ メでは装置と して正常に稼働しません。 そこで通信が正常に行われているかを常時確認す

る必要があるわけですが、高速で複雑な通信内容を管理するのは骨のおれるこ とです。前記のサンプルプロ

グラムでのチェッ クサムも通信内容の確認の1つで、 1キャラ ク ター受信のたびにそのチェッ クサムを計算

してチェ ッ クサムデータ と照合してデータが正常かど うかを確認しています。 このよ うにプログラムによ

るチェ ッ クのほか、 MPC-684ではRSE(n)関数によ り RS-232の受信状態を監視してソフ ト上で再送信要求な

どのエラー処理を行う こ とができます。

■RSE(n)関数

n=チャンネルナンバー　0または2

返される値 内容 エラー原因と して考えられるもの

0

1

2

4

8

正常終了

パリティーエラー

オーバーラン

フレーミ ングエラー

ブレークコード検出

ノ イズの混入

通信速度が速すぎる （ボーレート不一致）

ケーブル不良 ・初期化異常

異常データ

次のプログラムはMPC-684同士を接続し簡単な文字列を送受信したものですが、CNFGでの初期化に誤りが

あ りエラーと して検出されました。 エラーが発生した場合はCNFGコマンドで初期化し直します。

送信側

  CNFG#2 "9600b8pns1NONE" ’RS-232C CH2 初期化
   TIME 50
  PRINT#2 "ABC" CHR$(&HD)
  END

受信側

*LOOP
  CNFG#2 "9600b7pns1NONE" ’データビット長が違う
*LOOP1
  INPUT#2 A$
  SELECT_CASE RSE(2) ’RS-232C CH2チェック
    CASE 0 : PRINT A$ : GOTO *LOOP1
    CASE 1 : PRINT "ﾊﾟﾘﾃｨｰｴﾗｰ" : GOTO *LOOP ’エラーが起きたら
    CASE 2 : PRINT "ｵｰﾊﾞｰﾗﾝ" : GOTO *LOOP ’CNFGで初期化
    CASE 4 : PRINT "ﾌﾘｰﾐﾝｸﾞｴﾗｰ" : GOTO *LOOP
    CASE 8 : PRINT "ﾌﾞﾚｰｸｺｰﾄﾞ 検出" 　: GOTO *LOOP
    CASE_ELSE  : PRINT "???": GOTO *LOOP
  END_SELECT
  END

  RUN
  ﾌﾚｰﾐﾝｸﾞｴﾗｰ

RSE(n)関数では受信時のチェッ クはできますが、送信時のチェッ クはできません。 また、実際にはRSE(n)が

頻繁に活躍するのでは困り ものです。 RS-232 のエラーに限らずト ラブルは元から絶たなければ根本的な解

決にはなり ません。特に、別電源のパソコンや画像処理装置などの外部機器との接続にはフレームグランド

の共通化による浮遊電圧の防止や配線の経路・長さ・結線の方法、そして通信フォーマッ ト といったハード ・

ソフ ト両面での仕様を考慮して、 場合によってはアイ ソレーターを使用するなどの対策が必要になるこ と

もあ り ます。

タッチパネル（デジタル社製）の使用例

グラフ ィカルユーザー I/Fと してタ ッチパネルが普及しています。MPC-684シ リーズではMBK-SH,MBK-RS

でMEWNET-FPプロ ト コルをサポート し、通信を意識するこ とな く I/O機器と同レベルでGPを制御するこ と

ができます。 次は、 RS-232通信例と してのGP制御です。

（MBK-SH,MBK-RSについては第７章ハード リ ファレンス、 製品別マニュアルをご覧下さい。）

※MBK-RSは、 RS-232 CH0で直接GPをサポートする機能です。

■RS-232の結線

3線式ク ロスで結線しました。

Butaface

CH2 CH1

表示器 MPC-684
PC

【機器接続図】

1 CD
2 RXD
3 TXD
4 DTR
5 SG
6 DSR
7 RTS
8 CTS
9 C1

1 SG
2 TXD1
3 RXD1
4 SG
5 MAN
6 P5
7 SG
8 TXD2
9 RXD2
10

PCMPC-684ｺﾈｸﾀJ1
1 FG
2 SD
3 RD
4 RS
5 CS
7 SG
20 ER

表示器

■RS-232の初期化

9600bps、 8ビッ ト 、 パ リ ティー無し、 ス ト ップビッ ト 1、 XON/OFF

MPCの初期化は  CNFG#2 "9600b8pns1XON"　です。表示器側の設定を確認して下さい。表示器とやり と り

するデータはすべてASCII文字列です。 たとえば0を送るには’ 0’ （&H30） にな り ます。

■MPCのRS-232送受信バッファの確認の方法

プログラムのデバッグ中にRS-232から送信した り受信した り したデータを確認したい時があ り ます。 それ

にはRSコマンドを使用します。 バッファの内容をASCIIコードで表示します。

#PRINT#2 CHR$(&H1B) "R00100001¥r"
#RS 2
ＲＳ受信
00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
00 00 00 00 00 00 00 00 00 1B<41 30 30 30 31 0D <EC>A0001<CR>
ＲＳ送信 "<"記号は残バッファ先頭
00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
00 00 00 00 00 1B 52 30 30 31 30 30 30 30 31 0D <EC>R00100001<CR>
#

この送受信バッファは前記CNFG#2コマンドでク リ アーされます。

■表示器の画面

スイ ッチの入力、メ ッセージ表示、テンキーの使用、画面切り替えを想定して2つの画面をつく り ました。画

面の編集はデジタル社の専用ソフ トで作成しました。

【画面１ 】

「START」、 「STOP」、 「TEACH」 のスイ ッチと、 生産個数を表示す

るためのカウンターと リ セッ ト スイ ッチを設けてあ り ます。 その

下には 「運転中」 と 「停止」 の文字を表示するよ うにします。

【画面２ 】

点番号の入力のためのテンキーと

正常な入力か否かを表示する メ ッ

セージを表示します。

■動　作

･ 「START」 を押すとMPCは自動運転を行います。「運転中」 の表示をして、カウンターの値を加算していき

ます。

･ 「STOP」 を押すと停止します。

･ 「TEACH」 を押すと表示が画面２に切り替わり ます。 テンキーで点番号の設定を行います。

･ 「テンキー」 テンキーの数字を押すと左側に入力した数字を表示し、「ENT」 キーで確定してMPCがそれを

現在点と します。 も しその値が1～200であれば 「点データが登録されました」、 それ以外なら 「点データ

異常です。 再入力して ください。」 を表示します。 正常な入力ならば画面１に戻り ます。 異常な入力なら

ば画面２のまま再入力になり ます。

【画面１の構成】

RはカウンターのRESET SW

START  STOP TEACH

R (ｶｳﾝﾀｰ)

運転中

T1 T2 T3

T4  N1

L1 L2

7 8 9

1 2 3

4 5 6

0
Enter

DelIns

.

+

-*/Num
Lock

点番号を入力して下さい。

Ｐ＝ (点番号)  

点データが登録されました。

　 　 　 　 または

点データ異常です再入力し

て下さい。

【画面２の構成】

テンキーで入力され

た数字は(点番号)に表

示されます(3桁)。

■表示器システムエリアのアドレス設定

【画面１ 】

表示器システムエ リ ア

スイ ッチの確認、 メ ッセージ表示、 ページ切り替えは全て表示器のシス

テムエ リ アを介して行います。 スイ ッチが押されたらそれに対応したシ

ステムエ リ アのビッ トが0から1となる様にします。MPCは通信でそれを

読みだしてスイ ッチの状態を知るこ とができます。 メ ッセージも割り当

てられたシステムエ リ アのビッ トが 0 から 1 になる と画面に表示される

様にしておきます。 MPCは通信でそのビッ ト を1にセッ トするのです。

｢START｣を押すとアドレス16の第0ビッ トが1とな り ます。

｢STOP｣を押すとアドレス16の第1ビッ トが1とな り ます。

｢TEACH｣を押すとアドレス16の第2ビッ トが2とな り ます。

(ｶｳﾝﾀｰﾘｾｯﾄ)を押すとアドレス19の第2ビッ トが1とな り ます。

これらのＳＷは押されている間だけ1にな り ます。

アドレス17の第0ビッ ト を1にする と 「運転中」 が表示されます。

アドレス17の第1ビッ ト を1にする と 「停止」 が表示されます。

【画面２ 】

表示器システムエ リ ア

テンキーから入力されたデータはENTキーで確定されます。「ENT」を押すとアドレス19の第0ビッ トが1とな

り ます。 また、 そのと きのデータがアドレス0020に入り ます。

アドレス17の第2ビッ ト を1にする と 「点データが登録･･」 が表示されます。

アドレス17の第3ビッ ト を1にする と 「点データ異常･･」 が表示されます。

■画面１のSWの状態を知るには。

リード コマンドでシステムエ リ アからの読みだしをします。 リード コマンドを送信する と対応するアドレ

スのデータが返ってきます。 読みだしのフォーマッ トは次の通りです。

<EC>Raaaadddd<CR> 
<EC>はESCコード(&H1B)、 Rは” R” (&H52)、 aaaaはアドレス、 ddddは読みだすデータの数です。 　MPCでは

PRINT#2 CHR$(&H1B) "R00100001¥r"

ＳＷ名称 タグネーム アドレス

｢ｽﾀｰﾄ｣ T1 001600

｢STOP｣ T2 001601

｢TEACH｣ T3 001602

(ｶｳﾝﾀｰﾘｾｯﾄ) T4 001901

表示文字 タグネーム アドレス

｢運転中｣ L1 001700

｢停止｣ L2 001701

ＳＷ名称 タグネーム アドレス

（テンキー） 001800 ライブラリーを使用。

「ＥＮＴ」 T1 001900 テンキーのＥＮＴキー

表示文字 タグネーム アドレス

（点番号） K1   0020 テンキーを押すとその値が表示される

「点データ異常・・」 L1 001702

「点データが登録・・」 L2 001703

アドレス 16　　　SWオンで 1
BIT 0  0or1 ｢START」
    1   〃 「STOP」
　　2   〃 「TEACH」
    3   〃
    4   〃
    5   〃
    6   〃
    7   〃

アドレス 17　　　
BIT 0  0or1　　
    1   〃　　　1で表示
　　2   〃 「点データ異常」
    3   〃 「点データが登録」
    4   〃
    5   〃
    6   〃
    7   〃

読みだし
<EC>R

書き込み
<EC>W

表示器システムエリア
【読み込み ・書き込みイ メージ】

を送り ます。 こ こで注意しなければならないのがアドレスは16進数で表すこ とです。 アドレス16(DEC)なら
ば10(HEX)です。 返って く るデータのフォーマッ トは　

<EC>Adddd<CR>

です。 <EC>はESCコード(&H1B)、 Aは” A” (&H41)、 ddddはデータです。 MPCでは

INPUT#2 A$ 

で読み込みます。INPUT文は<CR>まで一括で読み込みます。も し 「START」 SWが押されていれば<EC>A0001
<CR>と返ってきます。「STOP」 が押されていれば<EC>A0002<CR>，「TEACH」 ならば<EC>A0004<CR> です。
何もおされていなければ<EC>A0000<CR>です。 こ こで注意しなければならないこ とは先頭にESCコードが
入っているためこのままでは変数に変換するこ とが出来ません。PRINT A$と実行しても表示されません。そ
こで次の様にして変数に変換します。

INPUT#2 A$ ’データをA$にいれます。
STRCPY A$ B$ 2 ’A$の3文字以降をB$にコピーします。
DATA=VAL(B$) ’文字列B$を数字に変換します。

例えばこの変換で 「START」 が押された場合、 変数DATAは1にな り ます。

■画面１の文字表示をするには。

ラ イ ト コマンドでシステムエ リ アへの書き込みをします。 書き込みのフォーマッ トは次の通りです。

<EC>Waaaadddd<CR>

<EC>はESCコード、 Wは” W” (&H57)、 aaaaはアドレス、 ddddはデータです。 「運転中」 はアドレス17の第0
ビッ ト を1にすれば表示されますから、

PRINT#2 CHR$(&H1B) "W00110001¥r"

と します。 こ こでもアドレス指定は16進数です。 (DEC)17は （HEX） 11です。 「停止」 を表示するにも同様に

PRINT#2 CHR$(&H1B) "W00110002¥r"       です。

■カウントの表示

こ こでは実際に何かを数える とい う訳ではあ り ませんが、一定時間毎に加算していく様にしています。この
カウンターは表示器が加算して くれるのではなく、 MPCから送られて く るデータを表示するだけです。

  　　I1=0
　　FOR I=0 TO 9999 ’0～9999までを表示します。
　　　　TIME 50
　　　　　　TMP$=STR$(I1)
　　　　　　COUNT$=""
　　　　　　STRCNT=4-LEN(TMP$) ’4byteのデータにするため”0”を連結します。
　　　　DO WHILE STRCNT<>0 ’カウント値が”12”なら”0012”にします。
　　　　　　　　COUNT$=COUNT$+"0"
　　　　　　　　STRCNT=STRCNT-1
　　　　LOOP
　　　　　　COUNT$=COUNT$+TMP$
　　　　PRINT COUNT$
　　　　　COUNT$=CHR$(&H1B)+"W0019"+COUNT$+"¥r" ’ついでにESCコードも<CR>も連結
　　　　PRINT#2 COUNT$
　　　　　　I1=I1+1
　　　　GOSUB *COUNTER_RESET ’カウンターリセットボタン入力確認サブルーチン
　　　　GOSUB *READ ’「停止」SW確認サブルーチン
　　　　IF DATA==2 THEN : RETURN  : END_IF
　　NEXT I
　　RETURN　
*COUNTER_RESET
　　PRINT#2 CHR$(&H1B) "R00130001¥r" ’リセットボタンの入力確認
　　INPUT#2 A$
　　STRCPY A$ B$ 2
　　IF VAL(B$)<>2 THEN : RETURN  : END_IF
　　  I1=0 ’リセットが押されたらカウント変数を0クリアー
　　RETURN

■画面の切り替え

画面の切り替えはアドレス15にページ番号を書き込みます。

PRINT#2 CHR$(&H1B) "W000F0002¥r" ’2ページ（画面２）へ切り替え
PRINT#2 CHR$(&H1B) "W000F0001¥r" ’1ページ（画面１）へ切り替え

■画面２の操作

テンキーを使うにはテンキーが設定されているアドレス18の第0ビッ トに1をセッ ト してイネーブル状態に

します。

PRINT#2 CHR$(&H1B) "W00120001¥r"

テンキーで入力したデータはアドレス20に入っています。MPCは 「ENT」 キーの押されるのを待ってこの20

のデータを読み込みます。

（「ENT」キーの入力待ち）
DO 
  PRINT#2 CHR$(&H1B) "R00130001¥r" ’アドレス19の読み込み
  INPUT#2 A$
  STRCPY A$ B$ 2
LOOP UNTIL VAL(B$)<>0
点データ入力
PRINT#2 CHR$(&H1B) "R00140001¥r" ’アドレス20の読み込み 
INPUT#2 A$
STRCPY A$ B$ 2
POINT=VAL(B$) ’変数POINTにﾃﾝｷｰのﾃﾞｰﾀがはいります。

■画面２の文字表示

前記POINTの値が1～200ならば 「点データが登録されました。」 を表示します。それ以外の数値なら 「点デー

タ異常です。 再入力して下さい。」 を表示します。

PRINT#2 CHR$(&H1B) "W00110008¥r" ’登録
PRINT#2 CHR$(&H1B) "W00110004¥r" ’異常

■プログラム全体

前記プログラムをま とめたものです。（*TASK1、*TASK2はﾀﾞﾐｰです） このプログラムはスイ ッチの監視のた

めに常にタスク0が通信文でLOOPしています。実際の装置ではもっと複雑な制御が要求されるので、この方

法では不具合なこ とがあるかもしれません。表示器についているI/Oの機能を利用して、 スイ ッチが押され

た時だけ通信する様に工夫すれば良いと思いますが．．．。

  FORK 1 *TASK1
  FORK 2 *TASK2
  CNFG#2 "9600b8pns1XON"
  PRINT#2 CHR$(&H1B) "W000F0001¥r"
  GOSUB *CLS1
  DO 
    DO 
      GOSUB *READ
    LOOP UNTIL DATA<>0
    SELECT_CASE DATA
      CASE 1 : GOSUB *START
      CASE 4 : GOSUB *TEACH
    END_SELECT 
    DO 
      GOSUB *READ
    LOOP UNTIL DATA==0
  LOOP 
*READ
  PRINT#2 CHR$(&H1B) "R00100001¥r"
  INPUT#2 A$
  STRCPY A$ B$ 2
    DATA=VAL(B$)
  RETURN 
*START
  GOSUB *CLS1
  PRINT#2 CHR$(&H1B) "W00110001¥r"
  GOSUB *COUNT
  RETURN 
*TEACH
  PRINT#2 CHR$(&H1B) "W000F0002¥r"
  GOSUB *TENKEY_READ
  TIME 1000
  PRINT#2 CHR$(&H1B) "W000F0001¥r"
  RETURN 
*CLS1

  PRINT#2 CHR$(&H1B) "W00110000¥r"
  RETURN 
*COUNT
    I1=0
  FOR I=0 TO 9999
    TIME 50
      TMP$=STR$(I1)
      COUNT$=""
      STRCNT=4-LEN(TMP$)
    DO WHILE STRCNT<>0
        COUNT$=COUNT$+"0"
        STRCNT=STRCNT-1
    LOOP 
      COUNT$=COUNT$+TMP$
    PRINT COUNT$
      COUNT$=CHR$(&H1B)+"W0019"+COUNT$+"¥r"
    PRINT#2 COUNT$
      I1=I1+1
    GOSUB *COUNTER_RESET
    GOSUB *READ
    IF DATA==2 THEN : RETURN  : END_IF 
  NEXT I
  RETURN 
'
*COUNTER_RESET
  PRINT#2 CHR$(&H1B) "R00130001¥r"
  INPUT#2 A$
  STRCPY A$ B$ 2
  IF VAL(B$)<>2 THEN : RETURN  : END_IF 
    I1=0
  RETURN 
*TENKEY_READ
  PRINT#2 CHR$(&H1B) "W00120001¥r" 'TENKEY ENB
  DO 
    PRINT#2 CHR$(&H1B) "R00130001¥r" 'ENT KEY READ 
    INPUT#2 A$
    STRCPY A$ B$ 2
  LOOP UNTIL VAL(B$)<>0
  PRINT#2 CHR$(&H1B) "R00140001¥r" 'TEN KEY DATA READ
  INPUT#2 A$
  PRINT A$
  STRCPY A$ B$ 2
    POINT=VAL(B$)
  IF POINT<=0 OR POINT>200 THEN
    PRINT#2 CHR$(&H1B) "W00110004¥r"
    TIME 500
    PRINT#2 CHR$(&H1B) "W00110000¥r"
    GOTO *TENKEY_READ
    ELSE 
    PRINT#2 CHR$(&H1B) "W00110008¥r"
    TIME 500
    PRINT#2 CHR$(&H1B) "W00110000¥r"
  END_IF 
  RETURN 
*TASK1
  DO 
    FOR J=0 TO 23
      ON J
      TIME 50
      OFF J
      TIME 50
    NEXT J
  LOOP 
*TASK2
  DO 
    FOR K=23 TO 47
      ON K
      TIME 50
      OFF K
      TIME 50
    NEXT K
  LOOP 
3-31


	第３章　プログラム例
	I/O制御・マルチタスク
	MPCの構成と装置概要
	I/Oマップ

	I/Oのシンボル化
	タスクの生成
	P&Pのタスク
	コンベアーＡのタスク
	コンベアーＢのタスク
	装置全体のプログラム

	タスクの制御
	タスクの交通整理（セマフォ）
	タスクのFORK、QUIT、PAUSE

	パルス発生について
	ボードアドレスとマルチタスク
	最初の確認
	※注意

	簡単なパルス発生プログラム
	※注意 INSET_314について

	マルチタスクでパルス発生
	INSET_314(条件設定)について
	動作範囲設定(ソフトリミット)
	アラーム、リミット入力の読み取り
	INCHK_314について
	多軸同時制御と直線補間
	■RMVS,MOVSコマンド
	■RMVL,MOVLコマンド
	■原点復帰

	溜まりパルスのリセット
	カウンタ機能
	◆STPS AXIS Val
	◆X(-1)〜U(-1)
	◆X(-2)〜U(-2)
	注(1):
	注(2):

	条件停止
	■ソフト停止
	◆STOP AXIS CMND

	■ハード入力検出停止
	◆STOP AXIS cond


	予約定数
	■指定ｸﾞﾙｰﾌﾟ
	■状態監視ｸﾞﾙｰﾌﾟ ■停止条件ｸﾞﾙｰﾌﾟ

	レジスタ概要
	■読み取り用
	◆RR0:パルス、エラ−監視
	◆RR1:軸毎のエラーステータス1
	◆RR2:軸毎のエラーステータス2
	◆RR3:軸毎のエラーステータス3

	■設定用
	◆WR0:コマンドレジスタ
	◆WR1:モードレジスタ1
	◆WR2:モードレジスタ2
	◆WR3:モードレジスタ3


	S字加減速
	引きずり現象について
	応用例
	■一枚のMPG-314の軸をタスクごとに非同期で使用する場合
	■一枚のMPG-314の軸で三軸直線補間と残りの軸を他のタスクで制御
	■パルス発生 途中停止や速度変更
	入力を検出してソフトで停止させる
	条件により速度を変更する
	原点復帰

	■パルス発生 途中停止 ハード
	INPOSの有効 サーボオンエラーリセットなど

	■移動距離を規定しないパルス発生 (RMVC,PLSC)
	■エラーの扱い方


	RS-232のプログラム
	MPC同士での通信例
	■ＭＰＣ−Ａ（送信側）
	送信フォーマット
	概 要

	■ＭＰＣ−Ｂ（受信側）
	受信フォーマット
	概　要


	RS-232のエラー処理
	■RSE(n)関数
	送信側
	受信側


	タッチパネル（デジタル社製）の使用例
	■RS-232の結線
	■RS-232の初期化
	■MPCのRS-232送受信バッファの確認の方法
	■表示器の画面
	【画面１】
	【画面２】

	■動　作
	■表示器システムエリアのアドレス設定
	【画面１】
	【画面２】

	■画面１のSWの状態を知るには。
	■画面１の文字表示をするには。
	■カウントの表示
	■画面の切り替え
	■画面２の操作
	■画面２の文字表示
	■プログラム全体




